

## CS 536 Announcements for Wednesday, February 21, 2024

### Programming Assignment 2

- due Tuesday, February 20 – accepted until 11:59 pm Thursday
- see late policy on course website

### Midterm 1

- Thursday, February 29, 7:30 – 9 pm
- S429 Chemistry
- bring your student ID

### Last Time

- implementing ASTs

### Today

- Java CUP

### Next Week

- review for Midterm 1
- parsing


## Parser generators

Tools that take an SDT spec and build an AST

- YACC *Yet Another C Compiler* – creates a parser in C
- Java CUP *Constructor of Useful Parsers* – creates a parser in Java

Conceptually similar to JLex:

- Input: language rules + actions
- Output: Java code



# Java CUP

## parser.java

- **constructor** takes argument of type `Yylex` *← from scanner*
- **parse** method
  - if input correct, returns Symbol whose value field contains translation of root nonterm
  - if input incorrect, quits on first syntax error
- **uses output of JLex** *← base.jlex*
  - depends on scanner and `TokenVal` classes
  - sym.java defines the communication language *= define token names used by both JLex & JavaCUP*
- uses definitions of AST classes (in ast.java)

## Parts of Java CUP specification

### Grammar rules with actions.

```

expr ::= INTLITERAL
      | ID
      | expr PLUS expr
      | expr TIMES expr
      | LPAREN expr RPAREN
      ;
  
```

*← not shown (yet)*

### Terminal and nonterminal declarations:

```

terminal      INTLITERAL;
terminal      ID;
terminal      PLUS;
terminal      TIMES;
terminal      LPAREN;
terminal      RPAREN;
  
```

```

non terminal  expr;
  
```

### Precedence and associativity declarations:

```

precedence left PLUS;
precedence left TIMES;
  
```

*↑ associativity*

*can do;*  

```
precedence nonassoc LESS;
```

*order (in ~.cup) indicates precedence*

*low  
↓  
high*

## Java CUP Example

### Assume:

defined in ast.java

- Java class `ExpNode` with subclasses `IntLitNode`, `IdNode`, `PlusNode`, `TimesNode`
- `PlusNode` and `TimesNode` each have two children
- `IdNode` has a `String` field (for the identifier)
- `IntLitNode` has an `int` field (for the integer value)
- `INTLITERAL` token is represented by `IntLitTokenVal` class and has field `intVal`
- `ID` token is represented by `IdTokenVal` class and has field `idVal`

### Step 1: add types to terminals and nonterminals

defined in base.jlex

```
/*
 * Terminal declarations
 */
terminal INTLITERAL;
terminal ID;
terminal PLUS;
terminal TIMES;
terminal LPAREN;
terminal RPAREN;
```

Need type if we want to use value associated with token

terminal `IntLitTokenVal INTLITERAL;`  
terminal `IdTokenVal ID;`

↳ from scanner (base.jlex)

```
/*
 * Nonterminal declarations
 */
non terminal expr;
```

Type required for all nonterms

non terminal `ExpNode expr;`

↳ from ast.java

### Step 2: add precedences and associativities

```
/*
 * Precedence and associativity declarations
 */
precedence left PLUS;
precedence left TIMES;
```

## Java CUP Example (cont.)

### Step 3: add actions to CFG rules

```
/*
 * Grammar rules with actions
 */
expr ::= INTLITERAL : i ← type is IntLitTokenVal
      {
        RESULT = new IntLitNode(i.intVal);
      }
| ID : i
      {
        RESULT = new IdNode(i.idVal);
      }
| expr : e1 PLUS expr : e2
      {
        RESULT = new PlusNode(e1, e2);
      }
| expr : e1 TIMES expr : e2
      {
        RESULT = new TimesNode(e1, e2);
      }
| LPAREN expr : e RPAREN
      {
        RESULT = e;
      }
;
```

Subclasses  
of  
ExpNode


### General format

```
nonterm ::= rule1
          {
            // action for rule1
            RESULT = ... ;
          }
| rule2
          {
            RESULT = ... ;
          }
| ...
;
```


# Java CUP Example (cont.)

Input: 2 + 3



Parse tree w/ translation



Built by parser



Built by scanner



## Translating lists

### Example

idList → idList COMMA ID | ID

← left recursive

### Left-recursion or right-recursion?

- for top-down parsers must use right recursion  
left-recursion leads to infinite loop
- for Java CUP use left recursion  
↳ bottom-up parser



## Example

CFG: idList → idList COMMA ID | ID

Goal: the translation of an idList is a LinkedList of Strings

### Example

Input: x , y , z



## Example (cont.)

idList → idList COMMA ID  
| ID

Java CUP specification for this syntax-directed translation

Terminal and nonterminal declarations:

terminal ~~ID~~TokenVal ID;  
terminal COMMA;  
non terminal LinkedList<String> idList;

Grammar rules and actions:

```
idList ::= idList : L      COMMA      ID : i
        { :
          L.addLast(i.idVal);
          RESULT = L;
        : }
| ID : i
  { :
    LinkedList<String> L = new LinkedList<String>( );
    L.add(i.idVal);
    RESULT = L;
  : }
;
```

## Handling unary minus

```
/*
 * precedences and associativities of operators
 */
precedence left PLUS, MINUS;
precedence left TIMES, DIVIDE;
precedence nonassoc UMINUS;
```

*binary minus has lowest precedence*

*"phony" token (never returned by scanner)*

*Unary minus has highest precedence*

```
/*
 * grammar rules
 */
exp ::= . . .
    | MINUS exp:e
      { : RESULT = new UnaryMinusNode(e);
        : } %prec UMINUS
    | exp:e1 PLUS exp:e2
      { : RESULT = new PlusNode(e1, e2);
        : }
    | exp:e1 MINUS exp:e2
      { : RESULT = new MinusNode(e1, e2);
        : }
    . . .
;
```

*Precedence of a rule is that of the last token of the rule, unless assigned a specific precedence via *%prec**